
	Reason for change:
	1. Unexpected message should be handled by TTCN and verdict should be made.

	
	

	Summary of change:
	1. Add activate default procedure

Change 1
	Function name
	fl_NR_Common_Init

	Reason for change
	Unexpected message should be handled by TTCN and verdict should be made.

	Summary of change
	Add activate default procedure

	TTCN module
	NR_CellInfoInit.ttcn

	MCC160 Comment
	


Before Change:

	    function fl_NR_Common_Init(NR_Configuration_Type p_NR_Configuration,

                             template (value) NR_Frequencyf1tof4_Type p_NR_PrimaryFrequencyf1tof4,

                             FreqBandIndicatorNR  p_PrimaryFreqBandIndicatorNR,

                             template (value) NR_Frequencyf1tof4_Type p_NR_SecondaryFrequencyf1tof4,

                             FreqBandIndicatorNR  p_SecondaryFreqBandIndicatorNR,

                             NR_SysinfoCombination_Type p_NR_SysinfoCombination)

    runs on NR_BASE_PTC

  {

    // Initialise Ctrl

    vc_NR_Global.Ctrl.Default := null;

    vc_NR_Global.Ctrl.TestBody := false;

    vc_NR_Global.Ctrl.VerdictCtrl := normal;

    vc_NR_Global.Ctrl.HarqErrorHandling := ignore;

    // Initialise security parameters

    vc_NR_Global.Security := f_NR_Security_Init();

    //Initialise CellArray

    vc_NR_Global.CellArray := f_NR_CellArray_Init(p_NR_PrimaryFrequencyf1tof4, p_PrimaryFreqBandIndicatorNR, p_NR_SecondaryFrequencyf1tof4, p_SecondaryFreqBandIndicatorNR, p_NR_SysinfoCombination);

    //Initialise MobileInfo

    vc_NR_Global.MobileInfo := f_NR_MobileInfo_Init();

    vc_NR_Global.NR_Config := p_NR_Configuration;

  }

	


After Change:

	   function fl_NR_Common_Init(NR_Configuration_Type p_NR_Configuration,

                             template (value) NR_Frequencyf1tof4_Type p_NR_PrimaryFrequencyf1tof4,

                             FreqBandIndicatorNR  p_PrimaryFreqBandIndicatorNR,

                             template (value) NR_Frequencyf1tof4_Type p_NR_SecondaryFrequencyf1tof4,

                             FreqBandIndicatorNR  p_SecondaryFreqBandIndicatorNR,

                             NR_SysinfoCombination_Type p_NR_SysinfoCombination)

    runs on NR_BASE_PTC

  {

    // Initialise Ctrl

    vc_NR_Global.Ctrl.Default := null;

    vc_NR_Global.Ctrl.TestBody := false;

    vc_NR_Global.Ctrl.VerdictCtrl := normal;

    vc_NR_Global.Ctrl.HarqErrorHandling := ignore;

    // Initialise security parameters

    vc_NR_Global.Security := f_NR_Security_Init();

    //Initialise CellArray

    vc_NR_Global.CellArray := f_NR_CellArray_Init(p_NR_PrimaryFrequencyf1tof4, p_PrimaryFreqBandIndicatorNR, p_NR_SecondaryFrequencyf1tof4, p_SecondaryFreqBandIndicatorNR, p_NR_SysinfoCombination);

    //Initialise MobileInfo

    vc_NR_Global.MobileInfo := f_NR_MobileInfo_Init();

    vc_NR_Global.NR_Config := p_NR_Configuration;

    f_NR_BASE_ActivateDefault();       

  }


Change 2
	template name
	cr_NR_SRB_COMMON_IND_Any

	Reason for change
	SRB unexpected message should be handled.

	Summary of change
	Add a new template for SRB used for change3

	TTCN module
	NR_Component_Base.ttcn

	MCC160 Comment
	


New template:

	  template (present) NR_SRB_COMMON_IND  cr_NR_SRB_COMMON_IND_Any := ?;


	


Change 3
	Function name
	a_NR_BASE_StandardDefault

	Reason for change
	Unexpected SRB message should be handled.

	Summary of change
	Handle  SRB receiving message

	TTCN module
	NR_Component_Base.ttcn

	MCC160 Comment
	


Before Change:

	  altstep a_NR_BASE_StandardDefault() runs on NR_BASE_PTC

  {

    var NR_SYSTEM_IND v_SYSTEM_IND;

    var charstring v_Msg;

    [] SYS.receive (cr_NR_SYSTEM_CTRL_CNF_Any) {

      /* unexpected CNF shall always result in INCONC */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected receive event at port SYS");

    }

    [] SYSIND.receive (cr_NR_SYSTEM_IND_Any) -> value v_SYSTEM_IND {

      if (ischosen(v_SYSTEM_IND.Indication.HarqError)) {

        select (vc_NR_Global.Ctrl.HarqErrorHandling) {

          case (ignore) {

            repeat;

          }

          case else {

            // don't repeat but terminate test case

          }

        }

      }

      all timer.stop;

      if (ischosen(v_SYSTEM_IND.Indication.Error)) {

        /* error indicated by the system simulator shall always result in INCONC */

        f_SetVerdictInconc(__FILE__, __LINE__, "error indicated at port SYS");

      } else {

        /* L1/MAC indications: all of these indications occur only when being activated by TTCN before */

        if (ischosen(v_SYSTEM_IND.Indication.RachPreamble) or

            ischosen(v_SYSTEM_IND.Indication.SchedReq) or

            ischosen(v_SYSTEM_IND.Indication.UL_HARQ) or

            ischosen(v_SYSTEM_IND.Indication.HarqError)) {

          /* L1 indications:

           * at layer 1 transmission errors may occur; therefore in general this is an unspecific behaviour */

          f_SetVerdictInconc(__FILE__, __LINE__, "unexpected L1 indication at port SYSIND");

        } else {

          /* MAC indications:

           * BSR, C_RNTI, PHR: within the test body a FAIL shall be assigned */

          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected MAC indication at port SYSIND");

        }

      }

    }

    [] DRB.receive (cr_NR_DRB_COMMON_IND_Any) {

      /* unexpected data at the DRB port */

      all timer.stop;

      v_Msg := "unexpected data at port DRB";

      if (vc_NR_Global.Ctrl.VerdictCtrl == layer2) {         /* user plane data is part of the test purpose */

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, v_Msg);

      } else {                                                  /* user plane data is not part of the test purpose */

        f_SetVerdictInconc(__FILE__, __LINE__, v_Msg);

      }

    }

    [] any timer.timeout {

      /* Note: there is no way to distinguish local timers in the default behaviour

       *       => there can be either an INCONC or a FAIL in the default behaviour;

       *       => we assign an INCONC - when there shall be FAIL in case of a timeout

       *          this needs to be handled where the timer is declared */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected timeout");

    }

  }


	


After Change:

	  altstep a_NR_BASE_StandardDefault() runs on NR_BASE_PTC

  {

    var NR_SYSTEM_IND v_SYSTEM_IND;

    var NR_SRB_COMMON_IND v_SRB_COMMON_IND;
    var charstring v_Msg;

    [] SYS.receive (cr_NR_SYSTEM_CTRL_CNF_Any) {

      /* unexpected CNF shall always result in INCONC */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected receive event at port SYS");

    }

    [] SYSIND.receive (cr_NR_SYSTEM_IND_Any) -> value v_SYSTEM_IND {

      if (ischosen(v_SYSTEM_IND.Indication.HarqError)) {

        select (vc_NR_Global.Ctrl.HarqErrorHandling) {

          case (ignore) {

            repeat;

          }

          case else {

            // don't repeat but terminate test case

          }

        }

      }

      all timer.stop;

      if (ischosen(v_SYSTEM_IND.Indication.Error)) {

        /* error indicated by the system simulator shall always result in INCONC */

        f_SetVerdictInconc(__FILE__, __LINE__, "error indicated at port SYS");

      } else {

        /* L1/MAC indications: all of these indications occur only when being activated by TTCN before */

        if (ischosen(v_SYSTEM_IND.Indication.RachPreamble) or

            ischosen(v_SYSTEM_IND.Indication.SchedReq) or

            ischosen(v_SYSTEM_IND.Indication.UL_HARQ) or

            ischosen(v_SYSTEM_IND.Indication.HarqError)) {

          /* L1 indications:

           * at layer 1 transmission errors may occur; therefore in general this is an unspecific behaviour */

          f_SetVerdictInconc(__FILE__, __LINE__, "unexpected L1 indication at port SYSIND");

        } else {

          /* MAC indications:

           * BSR, C_RNTI, PHR: within the test body a FAIL shall be assigned */

          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected MAC indication at port SYSIND");

        }

      }

    }

    [] DRB.receive (cr_NR_DRB_COMMON_IND_Any) {

      /* unexpected data at the DRB port */

      all timer.stop;

      v_Msg := "unexpected data at port DRB";

      if (vc_NR_Global.Ctrl.VerdictCtrl == layer2) {         /* user plane data is part of the test purpose */

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, v_Msg);

      } else {                                                  /* user plane data is not part of the test purpose */

        f_SetVerdictInconc(__FILE__, __LINE__, v_Msg);

      }

    }

    [] SRB.receive (cr_NR_SRB_COMMON_IND_Any) -> value v_SRB_COMMON_IND {

      /* unexpected L3 signalling message */

      all timer.stop;

      if (not ispresent(v_SRB_COMMON_IND.Signalling.Rrc)) {

        /* unexpected NAS message received in ULInformationTransfer => shall cause a FAIL when received in the test body */

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected NAS PDU at port SRB");

      } else if (not ispresent(v_SRB_COMMON_IND.Signalling.Nas)) {

        /* unexpected RRC message received => shall cause a FAIL when received in the test body */

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected RRC PDU at port SRB");

      } else {

        /* unexpected RRC message with piggybacked NAS PDU received => shall cause a FAIL when received in the test body */

        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "unexpected RRC PDU with piggybacked NAS PDU at port SRB");

      }

    }
    [] any timer.timeout {

      /* Note: there is no way to distinguish local timers in the default behaviour

       *       => there can be either an INCONC or a FAIL in the default behaviour;

       *       => we assign an INCONC - when there shall be FAIL in case of a timeout

       *          this needs to be handled where the timer is declared */

      all timer.stop;

      f_SetVerdictInconc(__FILE__, __LINE__, "unexpected timeout");

    }

  }

	


